/*
 * Flashlib - QB9 common Flash library
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */


package com.qb9.flashlib.collections
{
	import com.qb9.flashlib.interfaces.IDisposable;
	import com.qb9.flashlib.utils.ArrayUtil;
	
	// Abstract class
	
	public class BaseCollection implements IDisposable
	{
		protected var items:Array = [];
		protected var _size:uint;
		
		public function BaseCollection(size:uint = uint.MAX_VALUE)
		{
			_size = size;
		}
		
		// Exposed items state
		
		public function get size():uint
		{
			return _size;
		}
		
		public function get empty():Boolean
		{
			return items.length === 0;
		}
		
		public function get full():Boolean
		{
			return items.length >= size;
		}

		public function toArray():Array
		{
			return items.concat();
		}

		// Internal helpers

		protected function hasItem(obj:Object):Boolean
		{
			return items.indexOf(obj) !== -1;
		}
		
		protected function addItemAt(obj:Object, index:int):void
		{
			items.splice(index, 0, obj);
		}
		
		protected function removeItem(obj:Object):Object
		{
			return ArrayUtil.removeElement(items, obj);
		}
		
		// Disposal
		
		public function dispose():void
		{
			items = null;
		}

	}
}